

Ejercicio 3.7

[71.14] Modelos y Optimización I
Curso 4
2C 2021

Alumno:	Grassano, Bruno
Número de padrón:	103855
Email:	bgrassano@fi.uba.ar

Índice

1. Enunciado	2
2. Análisis de la situación problemática	3
3. Objetivo	3
4. Hipótesis y supuestos	3
5. Definición de variables	3
6. Modelo de programación lineal	4
6.1. Funcional	4
6.2. Restricciones	4
7. Resolución por software	5
8. Informe de la solución óptima	8

1. Enunciado

Una empresa mayorista compra y vende 3 productos A, B y C. En este momento tiene en stock 50, 100 y 300 unidades respectivamente. El precio de venta de A varía con la cantidad vendida: si vende hasta 500 unidades, \$ 20 c/u, de 500 a 1000, \$ 18 y más de 1000, \$ 15.

Los precios de venta de B y C son de \$ 35 y \$ 40 respectivamente.

La cantidad a entregar de A debe ser la menor de las tres pero si la cantidad entregada de B es menor que la de C esta limitación no se toma en cuenta.

Los precios de compra son los siguientes:

- producto A: \$ 3 c/u
- producto B: \$ 4 c/u si la compra de producto C es menor de 2500 un. y \$ 3. si la compra de C es mayor de 2500 unidades.
- producto C: \$ 6 c/u si se compran hasta 1000 un., \$ 5 si se compran menos de 3000 un. y \$ 4 para compras mayores de 3000 unidades.

Las demandas máximas son de 2000, 3000, y 4000 unidades respectivamente

2. Análisis de la situación problemática

- El precio de venta de A varía según la cantidad vendida, es similar al 3.6 con las máquinas. (Algo similar se ve con la compra del producto B y C) (La diferencia está que en este caso este precio no se paga una vez, se paga por cada unidad)
- Se tienen restricciones bivalentes.
- No se especifica el periodo.

3. Objetivo

Determinar la cantidad de cada tipo de producto a vender para maximizar las ganancias durante un periodo.

4. Hipótesis y supuestos

1. Todo lo puesto a la venta se vende. (Certeza)
2. Los precios son estables y exactos. (Certeza)
3. Se pueden hacer encargos y ventas de unidades (no es necesario que sea de a 10/20/...) (Certeza)
4. Los productos de un mismo tipo son indistinguibles. (Certeza)
5. Los productos no vienen con fallas. (Certeza)
6. No hay stock final. (Certeza)
7. El stock inicial está en buenas condiciones. (Certeza)
8. Los productos en stock son indistinguibles de los comprados. (Certeza)
9. Los productos en stock no tienen costos, estos ya fueron pagados en otro periodo. (Certeza)
10. No hay otros costos asociados al proceso. (Certeza)

5. Definición de variables

**Con tipos y unidades*

- PA_i : Indica la cantidad a vender de A dentro del rango de precio i . $i = 1, 2, 3$ (unidad/periodo)(Entera)
- Pj : Indica la cantidad a vender de j . $j = B, C$ (unidad/periodo)(Entera)
- CA : Indica la cantidad a comprar de A. (unidad/periodo)(Entera)
- Cj_i : Indica la cantidad a comprar de j dentro del rango de precio i . $i = 1, 2, 3$, $j = B, C$ (unidad/periodo)(Entera)
- YA_i : Indica si se usa el rango i de venta del producto A. $i = 1, 2, 3$ (Bivalente) (Vale 1 si está en el rango)
- Yj_i : Indica si se usa el rango i de compra del producto j . $i = 1, 2, 3$, $j = B, C$ (Bivalente) (Vale 1 si está en el rango) (El 3 de la i solo para C)
- YBC : Indica si la cantidad de B es menor que C. (Bivalente) (Vale 1 si es menor)

6. Modelo de programación lineal

**Indicando en cada restricción o grupo de restricciones la función que cumplen.*

6.1. Funcional

Buscamos maximizar la ganancia.

$$\max(20PA_1 + 18PA_2 + 15PA_3 + 35PB + 40PC - 3CA - 4CB_1 - 3CB_2 - 6CC_1 - 5CC_2 - 4CC_3)$$

6.2. Restricciones

Empezamos planteando las demandas máximas que tenemos.

- $PA_i \leq 2000$ Con $i = 1, 2, 3$
- $PB \leq 3000$
- $PC \leq 4000$

Planteamos las relaciones entre lo que se compra y lo que se vende.

- $CA + 50 = \sum_{i=1}^3 PA_i$
- $\sum_{i=1}^2 CB_i + 100 = PB$
- $\sum_{i=1}^3 CC_i + 300 = PC$

Planteamos el precio de venta de A.

- $1YA_1 \leq PA_1 \leq 499YA_1$
- $500YA_2 \leq PA_2 \leq 1000YA_2$
- $1001YA_3 \leq PA_3 \leq MYA_3$
- $\sum_{i=1}^3 YA_i = 1$

Varia el precio de compra de B.

- $1YB_1 \leq CC_1 + CC_2 + CC_3 \leq 2499YB_1 + MYB_2$
- $2500YB_2 \leq CC_1 + CC_2 + CC_3 \leq MYB_2 + MYB_1$
- $CB_1 \leq MYB_1$
- $CB_2 \leq MYB_2$
- $\sum_{i=1}^2 YB_i = 1$

Varia el precio de compra de C.

- $1YC_1 \leq CC_1 \leq 1000YC_1$
- $1001YC_2 \leq CC_2 \leq 2999YC_2$
- $3000YC_3 \leq CC_3 \leq MYC_3$
- $\sum_{i=1}^3 YC_i = 1$

Nos queda que la cantidad entregada de A debe ser menor que las 3, pero que si B es menor que C no se tiene en cuenta.

- $PA_1 + PA_2 + PA_3 \leq PB + MYBC$
- $PA_1 + PA_2 + PA_3 \leq PC + MYBC$
- $PC - MYBC \leq PB \leq PC + M(1 - YBC)$

7. Resolución por software

El modelo:

```

set Rango3;
set Rango2;

param M;
param preciosVentaA{i in Rango3};
param preciosCompraB{i in Rango2};
param preciosCompraC{i in Rango3};

var PA{i in Rango3} >=0 integer;
var PB >= 0 integer;
var PC >= 0 integer;
var CA >= 0 integer;
var CB{i in Rango2} >=0 integer;
var CC{i in Rango3} >=0 integer;
var YA{i in Rango3} >=0 binary;
var YB{i in Rango2} >=0 binary;
var YC{i in Rango3} >=0 binary;
var YBC >=0 binary;

maximize z: sum{i in Rango3} PA[i] * preciosVentaA[i] + 35 * PB + 40 * PC
- 3 * CA - (sum{i in Rango2} CB[i] * preciosCompraB[i])
- (sum{i in Rango3} CC[i] * preciosCompraC[i]);

s.t. limA1: PA[1] <= 2000;
s.t. limA2: PA[2] <= 2000;
s.t. limA3: PA[3] <= 2000;
s.t. limB: PB <= 3000;
s.t. limC: PC <= 4000;

s.t. compVentA: CA + 50 = sum{i in Rango3} PA[i];
s.t. compVentB: sum{i in Rango2} CB[i] + 100 = PB;
s.t. compVentC: sum{i in Rango3} CC[i] + 300 = PC;

s.t. rango1A1: YA[1] <= PA[1];
s.t. rango1A2: PA[1] <= 499 * YA[1];
s.t. rango2A1: 500 * YA[2] <= PA[2];
s.t. rango2A2: PA[2] <= 1000 * YA[2];
s.t. rango3A1: 1001 * YA[3] <= PA[3];
s.t. rango3A2: PA[3] <= M * YA[3];
s.t. limRangosA: sum{i in Rango3} YA[i] = 1;

s.t. rango1B1: YB[1] <= sum{i in Rango3} CC[i];
s.t. rango1B2: sum{i in Rango3} CC[i] <= 2499 * YB[1] + M * YB[2];
s.t. rango2B1: 2500 * YB[1] <= sum{i in Rango3} CC[i];
s.t. rango2B2: sum{i in Rango3} CC[i] <= M * YB[1] + M * YB[2];
s.t. opcionB1: CB[1] <= M * YB[1];
s.t. opcionB2: CB[2] <= M * YB[2];
s.t. limRangosB: sum{i in Rango2} YB[i] = 1;

```

```
s.t. rango1C1: YC[1] <= CC[1];
s.t. rango1C2: CC[1] <= 1000 * YC[1];
s.t. rango2C1: 1001 * YC[2] <= CC[2];
s.t. rango2C2: CC[2] <= 2999 * YC[2];
s.t. rango3C1: 3000 * YC[3] <= CC[3];
s.t. rango3C2: CC[3] <= M * YC[3];
s.t. limRangosC: sum{i in Rango3} YC[i] = 1;

s.t. limCantAB: sum{i in Rango3} PA[i] <= PB + M * YBC;
s.t. limCantAC: sum{i in Rango3} PA[i] <= PC + M * YBC;
s.t. menor1: PC - M * YBC <= PB;
s.t. menor2: PB <= PC + M * (1 - YBC);

data;
param M := 10000;

set Rango3 := 1 2 3;
set Rango2 := 1 2;

param preciosVentaA :=
1 20
2 18
3 15;

param preciosCompraB :=
1 4
2 3;

param preciosCompraC :=
1 6
2 5
3 4;

end;
```

Los resultados:

Problem: 3
 Rows: 34
 Columns: 20 (20 integer, 9 binary)
 Non-zeros: 97
 Status: INTEGER OPTIMAL
 Objective: z = 265650 (MAXimum)

No.	Row name	Activity	Lower bound	Upper bound
1	z	265650		
2	limA1	0		2000
3	limA2	0		2000
4	limA3	2000		2000
5	limB	3000		3000
6	limC	4000		4000
7	compVentA	-50	-50	=
8	compVentB	-100	-100	=
9	compVentC	-300	-300	=
10	rango1A1	0		-0
11	rango1A2	0		-0
12	rango2A1	0		-0
13	rango2A2	0		-0
14	rango3A1	-999		-0
15	rango3A2	-8000		-0
16	limRangosA	1	1	=
17	rango1B1	-3700		-0
18	rango1B2	-6300		-0
19	rango2B1	-3700		-0
20	rango2B2	-6300		-0
21	opcionB1	0		-0
22	opcionB2	-7100		-0
23	limRangosB	1	1	=
24	rango1C1	0		-0
25	rango1C2	0		-0
26	rango2C1	0		-0
27	rango2C2	0		-0
28	rango3C1	-700		-0
29	rango3C2	-6300		-0
30	limRangosC	1	1	=
31	limCantAB	-11000		-0
32	limCantAC	-12000		-0
33	menor1	-9000		-0
34	menor2	9000		10000

No.	Column name	Activity	Lower bound	Upper bound
1	PA[1]	*	0	0
2	PA[2]	*	0	0
3	PA[3]	*	2000	0
4	PB	*	3000	0
5	PC	*	4000	0
6	CA	*	1950	0

7	CB[1]	*	0	0	
8	CB[2]	*	2900	0	
9	CC[1]	*	0	0	
10	CC[2]	*	0	0	
11	CC[3]	*	3700	0	
12	YA[1]	*	0	0	1
13	YA[2]	*	0	0	1
14	YA[3]	*	1	0	1
15	YB[1]	*	0	0	1
16	YB[2]	*	1	0	1
17	YC[1]	*	0	0	1
18	YC[2]	*	0	0	1
19	YC[3]	*	1	0	1
20	YBC	*	1	0	1

Integer feasibility conditions:

KKT.PE: max.abs.err = 0.00e+00 on row 0
max.rel.err = 0.00e+00 on row 0
High quality

KKT.PB: max.abs.err = 0.00e+00 on row 0
max.rel.err = 0.00e+00 on row 0
High quality

End of output

8. Informe de la solución óptima

La forma para maximizar las ganancias es:

- vender 2000 (\$15 c/u) unidades de A comprando 1950 (\$3 c/u).
- vender 3000 (\$35 c/u) unidades de B comprando 2900 (\$3 c/u).
- vender 4000 (\$40 c/u) unidades de C comprando 3700 (\$4 c/u).

De esta forma se obtiene una ganancia de \$265.650.